

Dynamic Dashboards Using SAS®

Kirk Paul Lafler, Software Intelligence Corporation

Abstract

Dynamic interactive visual displays known as dashboards are most effective when they show essential graphs, tables, statistics, and other information where data is the star. The first rule for creating an effective dashboard is to keep it simple. Striking a balance between content and style, a dashboard should be void of excessive clutter so as not to distract and obscure the information displayed. The second rule of effective dashboard design involves displaying data that meets one or more business or organizational objectives. To accomplish this, the elements in a dashboard should convey a format easily understood by its intended audience. Attendees learn how to create dynamic interactive user- and data-driven dashboards, graphical and table-driven dashboards, statistical dashboards, and drill-down dashboards with a purpose.

Introduction

In a world of big data where data repositories and the demand placed on them are growing at explosive levels, organizations are faced with a number of decisions related to their information requirements:

- 1) What are the best ways to handle large amounts of information?
- 2) How should analytical data be processed?
- 3) What are the choices for constructing the most effective information delivery mechanisms?
- 4) How should analytical data and results be displayed?

To help answer these and other questions, this paper explains what a dashboard is, the dashboard's elements, the do's and don'ts for constructing effective dashboards, dashboard design techniques, an investigation of the various types of dashboards, the merits and strengths of using the base-SAS® software to construct dashboards, and an illustration of a few dashboard examples along with the base-SAS code used in their construction.

Example Table

The dashboard examples displayed in this paper reference a Movies table consisting of a number of movies that I've viewed over the years. The Movies table consists of six columns: Title, Length, Category, Year, Studio, and Rating. Title, Category, Studio, and Rating are defined as character columns, and the Length and Year are defined as numeric columns. The data contained in the Movies table is illustrated below.

Movies Table

	Title	Length	Category	Year	Studio	Rating
1	Brave Heart	177	Action Adventure	1995	Paramount Pictures	R
2	Casablanca	103	Drama	1942	MGM / UA	PG
3	Christmas Vacation	97	Comedy	1989	Warner Brothers	PG-13
4	Coming to America	116	Comedy	1988	Paramount Pictures	R
5	Dracula	130	Horror	1993	Columbia TriStar	R
6	Dressed to Kill	105	Drama Mysteries	1980	Filmways Pictures	R
7	Forrest Gump	142	Drama	1994	Paramount Pictures	PG-13
8	Ghost	127	Drama Romance	1990	Paramount Pictures	PG-13
9	Jaws	125	Action Adventure	1975	Universal Studios	PG
10	Jurassic Park	127	Action	1993	Universal Pictures	PG-13
11	Lethal Weapon	110	Action Cops & Robber	1987	Warner Brothers	R
12	Michael	106	Drama	1997	Warner Brothers	PG-13
13	National Lampoon's Vacation	98	Comedy	1983	Warner Brothers	PG-13
14	Poltergeist	115	Horror	1982	MGM / UA	PG
15	Rocky	120	Action Adventure	1976	MGM / UA	PG
16	Scarface	170	Action Cops & Robber	1983	Universal Studios	R
17	Silence of the Lambs	118	Drama Suspense	1991	Orion	R
18	Star Wars	124	Action Sci-Fi	1977	Lucas Film Ltd	PG
19	The Hunt for Red October	135	Action Adventure	1989	Paramount Pictures	PG
20	The Terminator	108	Action Sci-Fi	1984	Live Entertainment	R
21	The Wizard of Oz	101	Adventure	1939	MGM / UA	G
22	Titanic	194	Drama Romance	1997	Paramount Pictures	PG-13

“Brief” History of Dashboards

In the world of information technology, a dashboard serves as a user interface to organize and display information visually in the simplest way possible. Dashboards originated in the 1970’s as decision support tools and systems that served management, operations, and organizational planning. In the 1980’s, dashboards came of age as executive information systems emphasizing graphical displays and simple user interfaces to assist with management decision making. In the 1990’s, dashboards experienced a growing interest with the rise of the Internet. As information technology and the Internet entered the 2000’s, vendors including SAS Institute, and others, offered high-end easy-to-use products for the development of comprehensive “custom” dashboards. The dashboards being built today offer users the ability to monitor key metrics, information summaries, and reports in a single easy-to-use user interface. As a result, dashboards are designed to alert users to key business issues that impact an organization’s tactics and strategies by facilitating improved decision making activities.

So exactly what is a dashboard? In the paper, “Building Your First Dashboard Using the SAS® 9 Business Intelligence Platform: A Tutorial,” by Gregory S. Nelson (2009), Nelson describes a dashboard as a visualization technique that provides an immediate view or snapshot of exactly where you are in a specific process relative to your stated goals and objectives. He adds that, Visual indicators, such as temperature gauges, traffic lights and speedometers, help give a real-world sense of present progress and assists in making decisions, adapting to current conditions or drilling into more detailed information. As a user interface, dashboards display performance indicators (PIs), key performance indicators (KPIs), and other relevant information.

Types of Dashboards

The first step in dashboard design is to understand the purpose and type of dashboard you will need. With three types of dashboard designs available, users are encouraged to select the dashboard type that best meets your needs. The following table describes the three types of enterprise dashboards and their purpose.

Dashboard Type	Purpose
Strategic Dashboards	Strategic dashboards provide executives and managers with visual information to determine and support goals and objectives within an organization. This type of dashboard facilitates monitoring an organization’s health, progress, performance, and areas where improvement can be made. There is typically no need for interactive features with this type of dashboard. Strategic dashboard examples include: Sales, Human Resources, Manufacturing, and Services.
Analytical Dashboards	Analytical dashboards provide users with visual information to help gain a better understanding with historical, present and future data; understand trends; allow comparisons to be made; and determine the type of adjustments that are needed. Analytical dashboards should allow interactive features such as drill-down capabilities, as needed, to access more detailed information. Dashboard examples include: obtaining real-time data and information, determining why some things are working and others are not, identifying patterns and opportunities with your data, and aligning strategic objectives with performance initiatives.
Operational Dashboards	Operational dashboards provide users with visual information to concentrate on performance monitoring and measurements, monitor the efficiency and effectiveness of their organization. There is typically a need to update information displayed in an operational dashboard frequently to make it relevant to the users’ needs. Dashboard examples include: improved understanding of performance, better focus and alignment, and faster and better decision making.

Dashboard Elements

In Malik Shadan’s (2007) paper, Elements for an Enterprise Dashboard, he mentions that there are basic and advanced characteristics specific to an enterprise dashboard. The basic characteristics encompass the acronym, SMART, and the advanced characteristics of an enterprise dashboard encompass the acronym, IMPACT. The elements associated with each acronym appear in the following tables.

SMART Basic Elements	
	Description
Synergetic	Synergize information in a single screen view.
Monitor KPIs	Display critical KPIs for effective decision making.
Accurate	Dashboard must be well tested and validated, and information must be accurate.
Responsive	Respond to user alerts and visual content to draw immediate attention to critical matters.
Timely	Display information that is real-time and right-time for effective decision making.

IMPACT Advanced Elements

	Description
Interactive	Allow user to drill-down and derive details, root causes and more.
More Data History	Allow users to review historical trends for any KPI.
Personalized	Display should be specific to each user's domain of responsibility, data restrictions, and privileges.
Analytical	Allow users to perform guided analysis, compare, contrast, and make analytical inferences.
Collaborative	Facilitate users' ability to exchange notes regarding observations on their dashboard.
Trackability	Allow each user to customize the metrics they would like to track.

Avoiding 13 Common Pitfalls in Dashboard Design

Successful dashboard design involves the transformation of quantitative data into meaningful and effective visual displays including graphs, maps, gauges and summary information. In his paper, "Common Pitfalls in Dashboard Design," Stephen Few (2006) proposes 13 common mistakes many make when designing dashboards. Instead of concentrating on what should be done when designing dashboards, Mr. Few's body of work espouses the most common mistakes along with detailed explanations to help educate current and future designers alike. I have listed the 13 common pitfalls from Mr. Few's seminal work, below, but readers are encouraged to read his entire paper, see References, for a complete perspective.

Stephen Few's 13 Common Pitfalls in Dashboard Design (cited from reference)

Pitfall	Description
Pitfall #1	Exceeding the Boundaries of a Single Screen
Pitfall #2	Supplying Inadequate Context for the Data
Pitfall #3	Displaying Excessive Detail or Precision
Pitfall #4	Expressing Measures Indirectly
Pitfall #5	Choosing Inappropriate Media of Display
Pitfall #6	Introducing Meaningless Variety
Pitfall #7	Using Poorly Designed Display Media
Pitfall #8	Encoding Quantitative Data Inaccurately
Pitfall #9	Arranging the Data Poorly
Pitfall #10	Ineffectively Highlighting What's Important
Pitfall #11	Cluttering the Screen with Useless Decoration
Pitfall #12	Misusing or Overusing Color
Pitfall #13	Designing an Unappealing Visual Display

Steps to Creating a Dynamic Dashboard using Base-SAS®

Follow these basic steps to successfully construct a dynamic dashboard using the Base-SAS software.

1. Connect to desired data sources using Libname statement.
2. Create user-defined formats containing URL links for dashboard and drill-down results.
3. Create Graphics, Bar Chart, Box Plot, Histogram, Pie, etc.
4. Produce Graph Template using PROC SGRENDER;
5. Produce Drill-down Detail Reports, Summary Reports, Statistics, Tables, etc.

References

- Few, Stephen (2006), "Common Pitfalls in Dashboard Design," Copyright 2006, ProClarity Corporation, Boise, ID, USA.
http://www.perceptualedge.com/articles/Whitepapers/Common_Pitfalls.pdf
- Lafler, Kirk Paul (2012), "Exploring the SAS Metadata DICTIONARY Tables and SASHELP Views," Proceedings of the 2012 South Central SAS Users Group Conference, Software Intelligence Corporation, Spring Valley, CA, USA. <http://www.scsug.org/wp-content/uploads/2012/11/Exploring-DICTIONARY-Tables-and-SASHELP-Views-SCSUG-2012.pdf>
- Lafler, Kirk Paul (2013), "Hands-On SAS® Macro Programming Tips and Techniques," Proceedings of the 2013 SAS Global Forum (SGF) Conference, Software Intelligence Corporation, Spring Valley, CA, USA.
<http://support.sas.com/resources/papers/proceedings13/146-2013.pdf>
- Lafler, Kirk Paul (2013), PROC SQL: Beyond the Basics Using SAS, Second Edition, SAS Institute Inc., Cary, NC, USA.
http://www.sas.com/store/prodBK_62432_en.html
- Malik, Shadan (2007), "Elements for an Enterprise Dashboard," idashboards.com.
<http://www.dashboardinsight.com/articles/digital-dashboards/fundamentals/elements-for-an-enterprise-dashboard.aspx>
- Nelson, Gregory S. (2009), "Building Your First Dashboard Using the SAS® 9 Business Intelligence Platform: A Tutorial," Proceedings of the 2009 SAS Global Forum (SGF) Conference, ThotWave Technologies, Cary, NC, USA.
<http://support.sas.com/resources/papers/proceedings09/028-2009.pdf>
- Overton, Stephen (2012), "Lost in Wonderland? Methodology for a Guided Drill-Through Analysis Out of the Rabbit Hole," Proceedings of the 2012 SAS Global Forum (SGF) Conference, Zencos Consulting, Cary, NC, USA.
<http://support.sas.com/resources/papers/proceedings12/020-2012.pdf>
- Sams, Scott (2013), "SAS® BI Dashboard: Interactive, Data-Driven Dashboard Applications Made Easy," Proceedings of the 2013 SAS Global Forum (SGF) Conference, SAS Institute Inc, Cary, NC, USA.
<http://support.sas.com/resources/papers/proceedings13/061-2013.pdf>
- Zdeb, Mike (2004), "Pop-Ups, Drill-Downs, and Animation", Proceedings of the 2004 SAS Users Group International (SUGI) Conference, University at Albany School of Public Health, Rensselaer, NY, USA.
<http://www2.sas.com/proceedings/sugi29/090-29.pdf>

Acknowledgments

The author would like to thank the SGF 2015 Hands-on Workshop Section Chair, for accepting my abstract and paper; Tyler Smith, SGF 2015 Conference Chair; and the SGF 2015 Conference and Executive Committees for organizing a great conference!

Trademarks Citations

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

Author Information

Kirk Paul Lafler is consultant and founder of Software Intelligence Corporation and has been using SAS since 1979. He is a SAS Certified Professional, provider of IT consulting services, trainer to SAS users around the world, mentor, and sasCommunity.org emeritus Advisory Board member. As the author of six books including Google® Search Complete (Odyssey Press. 2014); PROC SQL: Beyond the Basics Using SAS, Second Edition (SAS Press. 2013); PROC SQL: Beyond the Basics Using SAS (SAS Press. 2004); Kirk has written more than five hundred papers and articles, been an Invited speaker and trainer at four hundred-plus SAS International, regional, special-interest, local, and in-house user group conferences and meetings, and is the recipient of 23 "Best" contributed paper, hands-on workshop (HOW), and poster awards.

Comments and suggestions can be sent to:

Kirk Paul Lafler

Senior SAS® Consultant, Application Developer, Data Scientist, Trainer and Author
Software Intelligence Corporation

E-mail: KirkLafler@cs.com

LinkedIn: <http://www.linkedin.com/in/KirkPaulLafler>

Twitter: @sasNerd

Dashboard Examples

1. PROC PRINT Dashboard using PROC FORMAT

HTML Drill-down User-defined Format in PROC PRINT by RATING

Title	Rating
The Wizard of Oz	G-rated Movie
Casablanca	PG-rated Movie
Jaws	PG-rated Movie
Pottergeist	PG-rated Movie
Rocky	PG-rated Movie
Star Wars	PG-rated Movie
The Hunt for Red October	PG-rated Movie
Christmas Vacation	PG-13-rated Movie
Forrest Gump	PG-13-rated Movie
Ghost	PG-13-rated Movie
Jurassic Park	PG-13-rated Movie
Michael	PG-13-rated Movie
National Lampoon's Vacation	PG-13-rated Movie
Titanic	PG-13-rated Movie
Brave Heart	R-rated Movie
Coming to America	R-rated Movie
Dracula	R-rated Movie
Dressed to Kill	R-rated Movie
Lethal Weapon	R-rated Movie
Scarface	R-rated Movie
Silence of the Lambs	R-rated Movie
The Terminator	R-rated Movie

(Click the underlined text to drill down.)

G-rated Movies					
Title	Length	Category	Year	Studio	Rating
The Wizard of Oz	101	Adventure	1939	MGM / UA	G
N = 1					
PG-rated Movies					
Title	Length	Category	Year	Studio	Rating
Casablanca	103	Drama	1942	MGM / UA	PG
Jaws	126	Action Adventure	1975	Universal Studios	PG
Pottergeist	115	Horror	1982	MGM / UA	PG
Rocky	120	Action Adventure	1976	MGM / UA	PG
Star Wars	124	Action Sci-Fi	1977	Lucas Film Ltd	PG
The Hunt for Red October	135	Action Adventure	1989	Paramount Pictures	PG
N = 6					
PG-13-rated Movies					
Title	Length	Category	Year	Studio	Rating
Christmas Vacation	87	Comedy	1989	Warner Brothers	PG-13
Forrest Gump	142	Drama	1994	Paramount Pictures	PG-13
Ghost	127	Drama Romance	1990	Paramount Pictures	PG-13
Jurassic Park	127	Action	1993	Universal Pictures	PG-13
Michael	108	Drama	1997	Warner Brothers	PG-13
National Lampoon's Vacation	88	Comedy	1993	Warner Brothers	PG-13
Titanic	194	Drama Romance	1997	Paramount Pictures	PG-13
N = 7					
R-rated Movies					
Title	Length	Category	Year	Studio	Rating
Brave Heart	177	Action Adventure	1999	Paramount Pictures	R
Coming to America	116	Comedy	1988	Paramount Pictures	R
Dracula	120	Horror	1959	Columbia TriStar	R
Dressed to Kill	109	Drama Mystery	1980	Foremost Pictures	R
Lethal Weapon	110	Action Cop & Robber	1987	Warner Brothers	R
Scarface	176	Action Cop & Robber	1983	Universal Studios	R
Silence of the Lambs	118	Drama Suspense	1991	Orion	R
The Terminator	108	Action Sci-Fi	1984	Live Entertainment	R
N = 8					

Base-SAS Code:

```

*****
**** Program Name: Drill-down PROC FORMAT with PROC PRINT.SAS ****;
**** Purpose.....: Create and display a simple drill-down user interface using ****;
**** PROC FORMAT and PROC PRINT with the ODS HTML destination. ****;
**** ****;
**** Author.....: Kirk Paul Lafler, Software Intelligence Corporation ****;
**** Date Written: 01/25/2015 ****;
**** SAS Version.: SAS 9.2, 9.3, 9.4 ****;
**** Input Files.: Workshop Data ****;
**** Output Files: None ****;
**** Subroutines.: None ****;
**** User-defined Formats: $RATFMT. ****;
**** Macro Variables: None ****;
**** Includes....: None ****;
**** Modification History: ****;
**** 02/12/2015 KPL Added Header information. ****;
*****
LIBNAME mydata 'e:\workshops\workshop data' ;
*STEP 1 - Create BY-Group by Sorting Movies Dataset ;
PROC SORT DATA=mydata.movies
      OUT=work.sorted_movies ;
      BY rating category ;
RUN ;

/*****
/* Step 2 - Create User-defined Rating Format with PROC FORMAT */
/*****
PROC FORMAT ;
  VALUE $RATFMT
    'G'   = '

```

```

*Step 3 - Create Drill-down HTML File, Titles and Footnote ;
ODS html file="e:\Drill-down-PROC-FORMAT.htm"
      (title="Drill-down User-defined Format in PROC PRINT") ;
TITLE1 h=.2in color=blue
      "HTML Drill-down User-defined Format in PROC PRINT" ;
TITLE2 color=blue "by RATING" ;
FOOTNOTE COLOR=blue "(Click the underlined text to drill down.)" ;
*Create the PROC PRINT Output with User-defined Format ;
OPTIONS NODATE ;
PROC PRINT DATA=work.sorted_movies LABEL NOOBS ;
  VAR rating ;
  FORMAT rating $RATFMT. ;
RUN ;
ODS html close ;


*Step 4 - Create Drill-down Output for Each Rating Group with PROC PRINT ;
TITLE ;
FOOTNOTE ;
ODS html body="e:\Drill-down-G-Movies.htm" ;
PROC PRINT DATA=mydata.movies NOOBS N ;
  TITLE "G-rated Movies" ;
  WHERE UPCASE(rating) = "G" ;
RUN ;
ODS html close ;

. . . . .

ODS html body="e:\Drill-down-R-Movies.htm" ;
PROC PRINT DATA=mydata.movies NOOBS N ;
  TITLE "R-rated Movies" ;
  WHERE UPCASE(rating) = "R" ;
RUN ;
ODS html close ;

```

2. Bar Chart Dashboard



Base-SAS Code:

G-rated Movies					
Title	Length	Category	Year	Studio	Rating
The Wizard of Oz	101	Adventure	1939	MGM / UA	G
N = 1					
PG-rated Movies					
Title	Length	Category	Year	Studio	Rating
Casablanca	103	Drama	1942	MGM / UA	PG
Jaws	125	Action/Adventure	1975	Universal Studios	PG
Pateget	115	Horror	1982	MGM / UA	PG
Raidy	120	Action/Adventure	1976	MGM / UA	PG
Star Wars	124	Action/Sci-Fi	1977	Lucas Film Ltd	PG
The Hunt for Red October	135	Action/Adventure	1990	Paramount Pictures	PG
N = 8					
PG-13-rated Movies					
Title	Length	Category	Year	Studio	Rating
Christmas Vacation	87	Comedy	1989	Warner Brothers	PG-13
Forest Gump	142	Drama	1994	Paramount Pictures	PG-13
Ghost	127	Drama/Romance	1990	Paramount Pictures	PG-13
Jurassic Park	127	Action	1993	Universal Pictures	PG-13
Michael	108	Drama	1997	Warner Brothers	PG-13
Nelson: Lampoon's Vacation	88	Comedy	1983	Warner Brothers	PG-13
Tease	104	Drama/Romance	1997	Paramount Pictures	PG-13
N = 7					
R-rated Movies					
Title	Length	Category	Year	Studio	Rating
Brave Heart	177	Action/Adventure	1995	Paramount Pictures	R
Coming to America	118	Comedy	1988	Paramount Pictures	R
Dreula	130	Horror	1993	Columbia TriStar	R
Dressed to Kill	105	Drama/Mystery	1980	Filmways Pictures	R
Lethal Weapon	110	Action/Comedy/Robber	1987	Warner Brothers	R
Starfire	170	Action/Comedy/Robber	1983	Universal Studios	R
Stamp of the Lantia	118	Drama/Suspense	1991	Copa	R
The Terminator	108	Action/Sci-Fi	1984	Live Entertainment	R
N = 8					

Base-SAS Code:

```
*****;
**** Program Name: Drill-down with PROC SGRENDER and BarChart.SAS ****;
**** Purpose.....: Create and display a simple drill-down dashboard interface ****;
**** using a DATA step, PROC TEMPLATE, PROC SGRENDER with the ****;
**** ODS HTML destination. ****;
**** Author.....: Kirk Paul Lafler, Software Intelligence Corporation ****;
**** Date Written: 02/19/2015 ****;
**** SAS Version.: SAS 9.2, 9.3, 9.4 ****;
**** Input Files.: Movies Workshop Data ****;
**** Output Files: HTML Output (6 Files) ****;
**** SGRender-BarChart-Dashboard.htm ****;
**** SGRender-BarChart-G-Movies.htm ****;
**** SGRender-BarChart-PG-Movies.htm ****;
**** SGRender-BarChart-PG13-Movies.htm ****;
**** SGRender-BarChart-PG17-Movies.htm ****;
**** SGRender-BarChart-R-Movies.htm ****;
**** Subroutines.: None ****;
**** User-defined Formats: None ****;
**** Macro Variables: None ****;
**** Includes.....: None ****;
**** Modification History: ****;
**** 02/21/2015 KPL Added Header information. ****;
*****;
libname mydata 'e:\workshops\workshop data';
*STEP 1 - Order Movies Dataset in Ascending Order by Rating ;
proc sort data=mydata.movies out=work.sorted_movies ;
  by rating ;
run ;

*STEP 2 - Assign HTML Link to URLLINK Variable ;
data Classic_Movies;
  length URLLINK $50;
  max=200;
  set work.sorted_movies;
  if upcase(rating) = 'G' then URLLINK='e:\SGRender-BarChart-G-Movies.htm';
  else
  if upcase(rating) = 'PG' then URLLINK='e:\SGRender-BarChart-PG-Movies.htm';
  else
  if upcase(rating) = 'PG-13' then URLLINK='e:\SGRender-BarChart-PG13-Movies.htm';
  else
  if upcase(rating) = 'PG-17' then URLLINK='e:\SGRender-BarChart-PG17-Movies.htm';
  else
  if upcase(rating) = 'R' then URLLINK='e:\SGRender-BarChart-R-Movies.htm';
run;

*STEP 3 - Create BarChart Template Overlay with PROC TEMPLATE ;
proc template ;
  define statgraph BarChartDashboard ;
    begingraph ;
    entrytitle "Drill-down BarChart Dashboard" ;
    layout overlay / xaxisopts=(display=(label tickvalues line)) ;
      barchart x=Rating y=length / url=urllink
              dataskin=crisp
              datatransparency=0.3
              orient=vertical ;
    endlayout ;
  endgraph ;
end ;
run ;

*Step 4 - Produce Graphical Output from BarChart Template using PROC SGRENDER;
ods html file='SGRender-BarChart-Dashboard.htm' path='e:\' ;
ods graphics / reset imagemap=on width=7in height=5in
  imagename='SGRender-BarChart-Dashboard' ;
title1 "BarChart of Movies Data" ;
title2 "Rating by Movie Length" ;
proc sgrender data=Classic_Movies template=BarChartDashboard ;
run ;
title ;
ods html close ;
```

```


*STEP 5 - Create Drill-down Output for Each Rating Group with PROC PRINT ;
ODS html body="SGRender-BarChart-G-Movies.htm" path='e:\';
PROC PRINT DATA=classic_movies(DROP=urllink max) NOOBS N ;
  TITLE "G-rated Movies" ;
  WHERE UPCASE(rating) = "G" ;
RUN ;
ODS html close ;

. . . . .

ODS html body="SGRender-BarChart-R-Movies.htm" path='e:\';
PROC PRINT DATA=classic_movies(DROP=urllink max) NOOBS N ;
  TITLE "R-rated Movies" ;
  WHERE UPCASE(rating) = "R" ;
RUN ;
ODS html close ;

```

3. Pie / Bar Chart Dashboard



G-rated Movies						
Title	Length	Category	Year	Studio	Rating	
The Godfather						
PG-rated Movies						
Cassidian						
PG-13-rated Movies						
Jeep						
Pottergeist						
R-rated Movies						
Former G:						
Star Wars	121	Action Adventure	1977	Paramount Pictures	R	
The Hunt	115	Comedy	1980	Paramount Pictures	R	
Home	130	Horror	1983	Columbia TriStar	R	
National L	105	Drama Mysteries	1980	Filmways Pictures	R	
Theme	110	Action Cops & Robber	1987	Warner Brothers	R	
Letal Weapon	110	Action Cops & Robber	1987	Warner Brothers	R	
Starforce	170	Action Cops & Robber	1983	Universal Studios	R	
Silence of the Lambs	115	Drama Suspense	1991	Orion	R	
The Terminator	105	Action Sci-Fi	1984	Live Entertainment	R	

Action Movies						
Title	Length	Category	Year	Studio	Rating	
Jurassic						
Horror Movies						
Dracula	130	Horror	1993	Columbia TriStar	R	
Pottergeist	115	Horror	1982	MGM / UA	PG	

Base-SAS Code:

```

*****;
**** Program Name: 2-Cell Drill-down, PROC SGRENDER, Pie and Bar Chart #2.SAS *****;
**** Purpose.....: Create and display a simple drill-down dashboard with two *****;
****                  categorical variables using a DATA step, PROC TEMPLATE, Pie, *****;
****                  Bar Chart, and PROC SGRENDER with the ODS HTML destination. *****;
**** *****;
**** Author.....: Kirk Paul Lafler, Software Intelligence Corporation *****;
**** Date Written: 03/28/2015 *****;
**** SAS Version.: SAS 9.2, 9.3, 9.4 *****;
**** Input Files.: Workshop Data *****;
****                  Movies *****;
**** Output Files: HTML Output (17 Files) *****;
****                  PieChart-BarChart-Dashboard.htm (Qty 1) *****;
****                  PieChart-Rating-Movies.htm (Qty 5) *****;
****                  BarChart-Category-Movies.htm (Qty 11) *****;
**** Subroutines.: None *****;
**** User-defined Formats: None *****;
**** Macro Variables: None *****;
**** Includes.....: None *****;
**** Modification History: *****;
**** 03/28/2015 KPL Added Header information. *****;

```

```

*****;
libname mydata 'e:\workshops\workshop data';

*****;
* Step 1 - Assign PieChart URL= to URLRATING Variable ;
*****;
data Classic_Movies;
  length URLRATING URLCATEGORY $60;
  max=200;
  set mydata.movies;
  if upcase(rating) = 'G' then URLRATING='e:\PieChart-G-Movies.htm';
  else
  if upcase(rating) = 'PG' then URLRATING='e:\PieChart-PG-Movies.htm';
  else
  if upcase(rating) = 'PG-13' then URLRATING='e:\PieChart-PG13-Movies.htm';
  else
  if upcase(rating) = 'R' then URLRATING='e:\PieChart-R-Movies.htm';

*****;
* Step 1b - Assign BarChart URL= to URLCATEGORY Variable ;
*****;
  if upcase(category) = 'ACTION' then
    URLCATEGORY='e:\BarChart-Action-Movies.htm';
  else
  if upcase(category) = 'ACTION ADVENTURE' then
    URLCATEGORY='e:\BarChart-ActionAdv-Movies.htm';
  else
  . . . . .

  else
  if upcase(category) = 'DRAMA SUSPENSE' then
    URLCATEGORY='e:\BarChart-DramaSuspense-Movies.htm';
  else
  if upcase(category) = 'HORROR' then
    URLCATEGORY='e:\BarChart-Horror-Movies.htm';
run;

* Step 2 - Create PieChart / BarChart Template Overlay with PROC TEMPLATE ;
proc template ;
  define statgraph PieChartBarChartDashboard ;
    begingraph ;
      entrytitle "Drill-down PieChart/BarChart Dashboard" ;
      layout gridded / columns=1 ;
        layout lattice / columns=2 ;
          cell ;
            layout region ;
              piechart category=Rating / url=URLRATING dataskin=crisp datatransparency=0.3 ;
            endlayout ;
          endcell ;
          cell ;
            layout overlay / width=250px xaxisopts=(display=(label tickvalues line)) ;
              barchart x=Category y=length / url=URLCATEGORY dataskin=crisp
                datatransparency=0.3
                orient=vertical barwidth=0.8 ;
            endlayout ;
          endcell ;
        endlayout ;
      endlayout ;
    endgraph ;
  end ;
run ;
quit ;

*****;
* Step 3 - Produce Graphical Output from PieChart BarChart ;
* Template using PROC SGRENDER ;
*****;
ods html file='SGRender-PieChart-BarChart-Dashboard.htm'
  path='e:\' ;
ods graphics / reset imagemap=on width=8in height=5in
  imagename='PieChartBarChart-Dashboard' ;
title1 "PieChart/BarChart of Movies Data" ;
title2 "Movie Ratings/Categories" ;
proc sgrender data=Classic_Movies

```

```

        template=PieChartBarChartDashboard ;

run ;
quit ;
title ;
ods html close ;

***** ;
* Step 4a - Create Pie Chart Drill-down Output for Each Rating Group with ;
*       PROC PRINT ;
***** ;
ODS html body="PieChart-G-Movies.htm" path='e:\';
PROC PRINT DATA=classic_movies(DROP=URLRATING URLCATEGORY MAX) NOOBS N ;
    TITLE "G-rated Movies" ;
    WHERE UPCASE(rating) = "G" ;
RUN ;
ODS html close ;

. . . . .

ODS html body="PieChart-R-Movies.htm" path='e:\';
PROC PRINT DATA=classic_movies(DROP=URLRATING URLCATEGORY MAX) NOOBS N ;
    TITLE "R-rated Movies" ;
    WHERE UPCASE(rating) = "R" ;
RUN ;
ODS html close ;

***** ;
* Step 4b - Create Bar Chart Drill-down Output for Each Category Group with ;
*       PROC PRINT ;
***** ;
ODS html body="BarChart-Action-Movies.htm" path='e:\';
PROC PRINT DATA=classic_movies(DROP=URLRATING URLCATEGORY MAX) NOOBS N ;
    TITLE "Action Movies" ;
    WHERE UPCASE(category) = "ACTION" ;
RUN ;
ODS html close ;

. . . . .

ODS html body="BarChart-Horror-Movies.htm" path='e:\';
PROC PRINT DATA=classic_movies(DROP=URLRATING URLCATEGORY MAX) NOOBS N ;
    TITLE "Horror Movies" ;
    WHERE UPCASE(category) = "HORROR" ;
RUN ;
ODS html close ;

```